A Project Report on

Mathematica as a Tool for Solving Problems in General Relativity

by

Tathagata Karmakar
SB–1312043

under guidance of

Dr. Tapobrata Sarkar

Department of Physics
Indian Institute of Technology, Kanpur
July, 2016

CERTIFICATE

The report titled

Mathematica as a Tool for Solving Problems in General Relativity

duly completed by
Tathagata Karmakar
SB–1312043

Submitted by,	Under supervision of,
Tathagata Karmakar 03.08. 2016	T. Sarkar 31.8.16.
Tathagata Karmakar Junior Undergraduate Department of Physics IIT Kanpur	Dr. Tapobrata Sarkar Professor Department of Physics IIT Kanpur

Date: August 1, 2016
Place: Indian Institite of Technology, Kanpur

Abstract

General theory of relativity provides us a new concept of gravity. Existence of gravitational waves, bending of starlight, precession of planetory orbits can be predicted using this theory. Beside all these, it is indispensable while exploring physics near a black hole. But it is extremely difficult, if not impossible, to solve Einstein's equations exactly. That's why various computational tools like Mathematica are becoming more useful day by day. This project explores Fermi normal coordinates near a Kerr metric and how Mathematica can be used in efficient manner to explore physics in Kerr metric. The benefit of using Fermi normal coordinates will be emphasized through expression of tidal potential in that frame in Kerr metric.

Contents

List of Figures

Aim

The main objectives of the project are

- To learn the theory of general relativity beginning from the basics and then studying application in case of a massive spherically symmetric star.
- To study the construction of Fermi normal coordinates and their importance.
- To learn use of RGTC package in Mathematica for calculations in general relativity.
- To analyze the black hole tidal problem in Fermi normal coordinates with efficient algorithm in mathematica for the calculations.

Work

2.1 Part I: Studying general relativity

First and foremost requirement of the project was being familiar with general relativity. To fulfil this I had to study general raltivity begining from the basics till some advanced topics. For most of the part lecture notes of Matthias Blau were used [1]. Topics studied were:

I. Definition of tensor, tensor algebra, covariant derivatives of tensor, parallel transport, Fermi-Walker parallel transport [1, 2].

II. Definition of metric, geodesic equation, Christoffel symbols [1, 2].

III. Principle of minimal coupling, energy-momentum tensor(of a perfect fluid), energy-momentum tensor as a source of gravity [1, 5].

IV. Riemann curvature tensor and its properties, geodesic deviation equation, Lie derivatives, Killing vectors(basics) [1].

V. Raychaudhuri equation for timelike geodesic congruences, Raychaudhuri equation for affine Null geodesic congruences, Raychaudhuri equation for non-affinely parametrized Null geodesics [1].

VI. Einstein's equation, cosmological constant [1].

VII. Static, spherically symmetric metrics, Schwarzschild metric, Birkhoff's theorem, interior solution for a static star, Tolman-Oppenheimer-Volkoff equation [1].

VIII. Equation for shape of orbit in Schwarzschild geometry, timelike and null geodesics. Precession of perehelia of planetory orbits, bending of Light by a star [1].

IX. Schwarzschild radius, Schwarzschild black holes, Eddington-Finkelstein coordinates, event horizons, Kruskal-Szekeres coordinates, maximal extension of Schwarzschild spacetime [1].

X. Construction of Fermi normal coordinate system and Fermi normal coordinate as local inertial frame(Mikowski metric and vanishing of Christoffel symbols) [6].

2.2 Part II: Mathematica and RGTC package for calculations in general relativity

RGTC(Riemannian Geometry and Tensor Calculus) package for Mathematica can be found in `http://www.inp.demokritos.gr/~sbonano/RGTC/`. The file EDCRGTCcode.m should be kept in $Path of Mathematica. Command ≪EDCRGTCcode.m loads the package.

To use the package one needs to define coordinates as a list of symbols and a symmetric matrix as metric. Then command RGtensors[metric, coordinates] calculates the following(here U means upper index and d means lower i.e. ${\Gamma^{\mu}}_{\alpha\beta}$ is denoted as GUdd[[μ,α,β]])*:

- Metric gdd (also a input).
- Inverse metric gUU.
- Chrisoffel symbols.

*See `http://www.inp.demokritos.gr/~sbonano/RGTC/NEBX-RGTC.pdf`

- Riemann tensors Rdddd i.e. $R_{\mu\nu\rho\lambda}$.
- Riemann tensors RUddd i.e. $R^{\mu}{}_{\nu\rho\lambda}$.
- Ricci tensor Rdd.
- Scalar curvature R.
- Weyl tensor & Einstein tensors.

A sample run is shown below:

```
In[1]:=

In[2]:= (*Ishii, section III: COMPONENTS OF THE RIEMANN TENSOR FOR A KERR SPACETIME*)
        (*To be remembered that index 1 corresponds to temporal component and 2,
        3,4 corresponds to 1,2,3 indices in Ishii's paper*)
        << EDCRGTCcode.m

In[1]:= BLcoord = {t, r, θ, φ};(*Boyer-Lindquist coordinates*)

In[2]:= Σ = r^2 + a^2 Cos[θ]^2; ΔΔ = r^2 + a^2 - 2 M r; simpRules = TrigRules;
        (*Here Δ in Ishii's paper is written as ΔΔ*)assmp =
         ((2 r^2 + a^2 + a^2 Cos[2 θ]) ≥ 0) && (ΔΔ ≥ 0) && ( 2 Σ == 2 r^2 + a^2 + a^2 Cos[2 θ]) && (r >= 0);
        (*M and a denote mass and spin parameter respectively*)
```

$$\text{In[3]:= gBL} = \begin{pmatrix} -1+2\,M\,r/\Sigma & 0 & 0 & -2\,M\,r\,a\,\mathrm{Sin}[\theta]\string^2/\Sigma \\ 0 & \Sigma/\Delta\Delta & 0 & 0 \\ 0 & 0 & \Sigma & 0 \\ -2\,M\,r\,a\,\mathrm{Sin}[\theta]\string^2/\Sigma & 0 & 0 & ((r\string^2+a\string^2)\string^2-\Delta\Delta\,a\string^2\,\mathrm{Sin}[\theta]\string^2)\,\mathrm{Sin}[\theta]\string^2/\Sigma \end{pmatrix};$$

```
In[4]:= RGtensors[gBL, BLcoord];
```

$$\text{gdd} = \begin{pmatrix} -1+\frac{2\,M\,r}{r^2+a^2\,\mathrm{Cos}[\theta]^2} & 0 & 0 & -\frac{2\,a\,M\,r\,\mathrm{Sin}[\theta]^2}{r^2+a^2\,\mathrm{Cos}[\theta]^2} \\ 0 & \frac{r^2+a^2\,\mathrm{Cos}[\theta]^2}{a^2-2\,M\,r+r^2} & 0 & 0 \\ 0 & 0 & r^2+a^2\,\mathrm{Cos}[\theta]^2 & 0 \\ -\frac{2\,a\,M\,r\,\mathrm{Sin}[\theta]^2}{r^2+a^2\,\mathrm{Cos}[\theta]^2} & 0 & 0 & \frac{\mathrm{Sin}[\theta]^2\left((a^2+r^2)^2-a^2\,(a^2-2\,M\,r+r^2)\,\mathrm{Sin}[\theta]^2\right)}{r^2+a^2\,\mathrm{Cos}[\theta]^2} \end{pmatrix}$$

Figure 2.1: Inputs & Commands for Running RGTC Package

$$
\texttt{gdd} = \begin{pmatrix} -1+\frac{2\,M\,r}{r^2+a^2\,\mathrm{Cos}[\theta]^2} & 0 & 0 & -\frac{2\,a\,M\,r\,\mathrm{Sin}[\theta]^2}{r^2+a^2\,\mathrm{Cos}[\theta]^2} \\ 0 & \frac{r^2+a^2\,\mathrm{Cos}[\theta]^2}{a^2-2\,M\,r+r^2} & 0 & 0 \\ 0 & 0 & r^2+a^2\,\mathrm{Cos}[\theta]^2 & 0 \\ -\frac{2\,a\,M\,r\,\mathrm{Sin}[\theta]^2}{r^2+a^2\,\mathrm{Cos}[\theta]^2} & 0 & 0 & \frac{\mathrm{Sin}[\theta]^2\left(\left(a^2+r^2\right)^2-a^2\left(a^2-2\,M\,r+r^2\right)\mathrm{Sin}[\theta]^2\right)}{r^2+a^2\,\mathrm{Cos}[\theta]^2} \end{pmatrix}
$$

$$
\begin{aligned}
\texttt{LineElement} = {} & \frac{\left(r^2+a^2\,\mathrm{Cos}[\theta]^2\right)\,d[r]^2}{a^2-2\,M\,r+r^2}+\frac{\left(2\,M\,r-r^2-a^2\,\mathrm{Cos}[\theta]^2\right)\,d[t]^2}{r^2+a^2\,\mathrm{Cos}[\theta]^2}+ \\
& \left(r^2+a^2\,\mathrm{Cos}[\theta]^2\right)\,d[\theta]^2-\frac{4\,a\,M\,r\,d[t]\,d[\varphi]\,\mathrm{Sin}[\theta]^2}{r^2+a^2\,\mathrm{Cos}[\theta]^2}+\frac{1}{r^2+a^2\,\mathrm{Cos}[\theta]^2} \\
& d[\varphi]^2\,\mathrm{Sin}[\theta]^2\left(a^4+2\,a^2\,r^2+r^4-a^4\,\mathrm{Sin}[\theta]^2+2\,a^2\,M\,r\,\mathrm{Sin}[\theta]^2-a^2\,r^2\,\mathrm{Sin}[\theta]^2\right)
\end{aligned}
$$

$$
\begin{aligned}
\texttt{gUU} = {} & \Big\{\Big\{-\left(\left(r^2+a^2\,\mathrm{Cos}[\theta]^2\right)\left(a^4+2\,a^2\,r^2+r^4-a^4\,\mathrm{Sin}[\theta]^2+2\,a^2\,M\,r\,\mathrm{Sin}[\theta]^2-a^2\,r^2\,\mathrm{Sin}[\theta]^2\right)\right)\Big/ \\
& \quad \Big(-2\,a^4\,M\,r+a^4\,r^2-4\,a^2\,M\,r^3+2\,a^2\,r^4-2\,M\,r^5+r^6+a^6\,\mathrm{Cos}[\theta]^2+2\,a^4\,r^2\,\mathrm{Cos}[\theta]^2+ \\
& \qquad a^2\,r^4\,\mathrm{Cos}[\theta]^2+2\,a^4\,M\,r\,\mathrm{Sin}[\theta]^2-a^4\,r^2\,\mathrm{Sin}[\theta]^2+4\,a^2\,M\,r^3\,\mathrm{Sin}[\theta]^2-a^2\,r^4\,\mathrm{Sin}[\theta]^2- \\
& \qquad a^6\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta]^2+2\,a^4\,M\,r\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta]^2-a^4\,r^2\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta]^2\Big),\ 0,\ 0, \\
& \quad \left(2\,a\,M\,r\left(r^2+a^2\,\mathrm{Cos}[\theta]^2\right)\right)\Big/\Big(2\,a^4\,M\,r-a^4\,r^2+4\,a^2\,M\,r^3-2\,a^2\,r^4+2\,M\,r^5-r^6-a^6\,\mathrm{Cos}[\theta]^2- \\
& \qquad 2\,a^4\,r^2\,\mathrm{Cos}[\theta]^2-a^2\,r^4\,\mathrm{Cos}[\theta]^2-2\,a^4\,M\,r\,\mathrm{Sin}[\theta]^2+a^4\,r^2\,\mathrm{Sin}[\theta]^2-4\,a^2\,M\,r^3\,\mathrm{Sin}[\theta]^2+ \\
& \qquad a^2\,r^4\,\mathrm{Sin}[\theta]^2+a^6\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta]^2-2\,a^4\,M\,r\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta]^2+a^4\,r^2\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta]^2\Big)\Big\}, \\
& \left\{0,\ \frac{a^2-2\,M\,r+r^2}{r^2+a^2\,\mathrm{Cos}[\theta]^2},\ 0,\ 0\right\},\ \left\{0,\ 0,\ \frac{1}{r^2+a^2\,\mathrm{Cos}[\theta]^2},\ 0\right\}, \\
& \Big\{\left(2\,a\,M\,r\left(r^2+a^2\,\mathrm{Cos}[\theta]^2\right)\right)\Big/\Big(2\,a^4\,M\,r-a^4\,r^2+4\,a^2\,M\,r^3-2\,a^2\,r^4+2\,M\,r^5-r^6-a^6\,\mathrm{Cos}[\theta]^2- \\
& \qquad 2\,a^4\,r^2\,\mathrm{Cos}[\theta]^2-a^2\,r^4\,\mathrm{Cos}[\theta]^2-2\,a^4\,M\,r\,\mathrm{Sin}[\theta]^2+a^4\,r^2\,\mathrm{Sin}[\theta]^2-4\,a^2\,M\,r^3\,\mathrm{Sin}[\theta]^2+ \\
& \qquad a^2\,r^4\,\mathrm{Sin}[\theta]^2+a^6\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta]^2-2\,a^4\,M\,r\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta]^2+a^4\,r^2\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta]^2\Big), \\
& \quad 0,\ 0,\ \left(\left(2\,M\,r-r^2-a^2\,\mathrm{Cos}[\theta]^2\right)\left(r^2+a^2\,\mathrm{Cos}[\theta]^2\right)\mathrm{Csc}[\theta]^2\right)\Big/ \\
& \quad \Big(2\,a^4\,M\,r-a^4\,r^2+4\,a^2\,M\,r^3-2\,a^2\,r^4+2\,M\,r^5-r^6-a^6\,\mathrm{Cos}[\theta]^2-2\,a^4\,r^2\,\mathrm{Cos}[\theta]^2- \\
& \qquad a^2\,r^4\,\mathrm{Cos}[\theta]^2-2\,a^4\,M\,r\,\mathrm{Sin}[\theta]^2+a^4\,r^2\,\mathrm{Sin}[\theta]^2-4\,a^2\,M\,r^3\,\mathrm{Sin}[\theta]^2+a^2\,r^4\,\mathrm{Sin}[\theta]^2+ \\
& \qquad a^6\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta]^2-2\,a^4\,M\,r\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta]^2+a^4\,r^2\,\mathrm{Cos}[\theta]^2\,\mathrm{Sin}[\theta]^2\Big)\Big\}\Big\}
\end{aligned}
$$

```
gUU computed in 0.172 sec

Gamma computed in 0.125 sec

Riemann(dddd) computed in 1.063 sec

Riemann(Uddd) computed in 10.375 sec

Ricci computed in 2.609 sec

Weyl computed in 18.359 sec

Einstein computed in 5.625 sec

All tasks completed in 38.359 seconds
```

Figure 2.2: Outputs after Running RGTC Package

2.3 Part III: Kerr Spacetime Through Mathematica

This part involved first finding Riemann tensors and its derivatives in Fermi normal coordinates. Then calculating tidal potential upto fourth order terms. For most part assistance was taken from Ishii's paper [3].
First of all, Kerr metric in Boyer-Lindquist coordinate system is given as:

$$\begin{aligned} ds^2 = -(1-\frac{2Mr}{\Sigma})dt^2 - \frac{4Mra\sin^2\theta}{\Sigma}dtd\varphi \\ + \frac{(r^2+a^2)^2-\Delta a^2\sin^2\theta}{\Sigma}\sin^2\theta d\varphi^2 \\ + \frac{\Sigma}{\Delta}dr^2 + \Sigma d\theta^2 \end{aligned} \tag{2.1}$$

Where

$$\Sigma = r^2 + a^2\cos^2\theta, \qquad \Delta = r^2 + a^2 - 2Mr \tag{2.2}$$

M and a are mass and spin parameter respectively.

In Mathematica the commands are:

```
In[1]:=

In[2]:= (*Ishii, section III: COMPONENTS OF THE RIEMANN TENSOR FOR A KERR SPACETIME*)
        (*To be remembered that index 1 corresponds to temporal component and 2,
        3,4 corresponds to 1,2,3 indices in Ishii's paper*)
        << EDCRGTCcode.m

In[1]:= BLcoord = {t, r, θ, φ};(*Boyer-Lindquist coordinates*)

In[2]:= Σ = r^2 + a^2 Cos[θ]^2; ΔΔ = r^2 + a^2 - 2 M r; simpRules = TrigRules;
        (*Here Δ in Ishii's paper is written as ΔΔ*)assmp =
         ((2 r^2 + a^2 + a^2 Cos[2 θ]) ≥ 0) && (ΔΔ ≥ 0) && ( 2 Σ == 2 r^2 + a^2 + a^2 Cos[2 θ]) && (r >= 0);
        (*M and a denote mass and spin parameter respectively*)

In[3]:= gBL = ( -1 + 2 M r / Σ           0       0   -2 M r a Sin[θ]^2 / Σ
                0                        Σ / ΔΔ  0   0
                0                        0       Σ   0
                -2 M r a Sin[θ]^2 / Σ    0       0   ((r^2 + a^2)^2 - ΔΔ a^2 Sin[θ]^2) Sin[θ]^2 / Σ );

In[4]:= RGtensors[gBL, BLcoord];
```

$$\text{gdd} = \begin{pmatrix} -1+\frac{2\,M\,r}{r^2+a^2\,\text{Cos}[\theta]^2} & 0 & 0 & -\frac{2\,a\,M\,r\,\text{Sin}[\theta]^2}{r^2+a^2\,\text{Cos}[\theta]^2} \\ 0 & \frac{r^2+a^2\,\text{Cos}[\theta]^2}{a^2-2\,M\,r+r^2} & 0 & 0 \\ 0 & 0 & r^2+a^2\,\text{Cos}[\theta]^2 & 0 \\ -\frac{2\,a\,M\,r\,\text{Sin}[\theta]^2}{r^2+a^2\,\text{Cos}[\theta]^2} & 0 & 0 & \frac{\text{Sin}[\theta]^2\,((a^2+r^2)^2-a^2\,(a^2-2\,M\,r+r^2)\,\text{Sin}[\theta]^2)}{r^2+a^2\,\text{Cos}[\theta]^2} \end{pmatrix}$$

Figure 2.3: Inputs & Commands for Kerr spacetime

Standard tetrad is given as(see [3] eqns 74, 75, 76, 76):

```
In[5]:= (*Standard Tetrad*)
In[6]:= evec0 = {Sqrt[ΔΔ / Σ], 0, 0, -a Sin[θ]^2 Sqrt[ΔΔ / Σ]};
In[7]:= evec1 = {0, Sqrt[Σ / ΔΔ], 0, 0};
In[8]:= evec2 = {0, 0, Sqrt[Σ], 0};
In[9]:= evec3 = {-a Sin[θ] / Sqrt[Σ], 0, 0, (r^2 + a^2) Sin[θ] / Sqrt[Σ]};
In[10]:= (*To see they indeed are orthonormal*)
         {evec0.gUU.evec0, evec0.gUU.evec1, evec0.gUU.evec2, evec0.gUU.evec3} // FullSimplify
Out[10]= {-1, 0, 0, 0}
In[11]:= {evec1.gUU.evec0, evec1.gUU.evec1, evec1.gUU.evec2, evec1.gUU.evec3} // FullSimplify
Out[11]= {0, 1, 0, 0}
In[12]:= {evec2.gUU.evec0, evec2.gUU.evec1, evec2.gUU.evec2, evec2.gUU.evec3} // FullSimplify
Out[12]= {0, 0, 1, 0}
In[13]:= {evec3.gUU.evec0, evec3.gUU.evec1, evec3.gUU.evec2, evec3.gUU.evec3} // FullSimplify
Out[13]= {0, 0, 0, 1}
```

Figure 2.4: Standard Tetrad

The tetrad components ($e^{(a)}_{\mu}$) are written as a matrix named jac. The matrix inverse is ijac ($e^{\mu}_{(a)}$). R1dddd is full simplified form of Rdddd (for computations with R1dddd will take lesser time though computation of R1dddd might take 2-3 minutes). A picture is shown in the next page.

```
In[14]:= jac = {evec0, evec1, evec2, evec3};(*Writing tetrad components e^(a)_μ as matrix jac*)
In[15]:= ijac = FullSimplify[Inverse[jac], assmp];(*ijac consists of e_(a)^μ*)
In[16]:=

In[17]:= (*R1dddd is Simplified version of Rdddd*)
         R1dddd = Table[FullSimplify[Rdddd[[i, j, k, l]], assmp],
            {i, 1, 4}, {j, 1, 4}, {k, 1, 4}, {l, 1, 4}];
```

Figure 2.5: Matrix and inverse matrix of tetrad components

Calculation of components of Riemann tensor in tetrad frame is given on next page:

```
In[20]:= (*Below combpos creates all even combinations of first four inputs
          and combneg creates odd combinations*)combpos[i_, j_, k_, l_, m___] :=
           {{i, j, k, l, m}, {j, i, l, k, m}, {k, l, i, j, m}, {l, k, j, i, m}};

In[21]:= combneg[i_, j_, k_, l_, m___] :=
           {{i, j, l, k, m}, {j, i, k, l, m}, {l, k, i, j, m}, {k, l, j, i, m}};

In[22]:= A = {};(*A is to be filled with independent combinations of type i-j-k-l*)

In[23]:= (*This function returns true if none of the combinations of input is already in A*)
         check[i_, j_, k_, l_] :=
          check[i, j, k, l] = (! MemberQ[A, {i, j, l, k}]) && (! MemberQ[A, {j, i, k, l}]) &&
            (! MemberQ[A, {j, i, l, k}]) && (! MemberQ[A, {l, k, i, j}]) &&
            (! MemberQ[A, {k, l, i, j}]) && (! MemberQ[A, {k, l, j, i}]) && (! MemberQ[A, {l, k, j, i}])

In[24]:= Do[A = If[check[i, j, k, l], Append[A, {i, j, k, l}], A],
           {i, 1, 4}, {j, 1, 4}, {k, 1, 4}, {l, 1, 4}];

In[25]:= (*Calculation of tetrad components of Rdddd*)
         calR[i_, j_, k_, l_] := calR[i, j, k, l] = If[i == j || k == l, 0,
             FullSimplify[Sum[R1dddd[[i1, i2, i3, i4]] ijac[[i1, i]] ijac[[i2, j]] ijac[[i3, k]]
                ijac[[i4, l]], {i1, 1, 4}, {i2, 1, 4}, {i3, 1, 4}, {i4, 1, 4}], assmp]];

In[26]:= Rlist = {};

In[27]:= Do[Rlist = Append[Rlist, calR[A[[i]] /. List → Sequence]], {i, Length[A]}];

In[28]:= Rtetrad[i_, j_, k_, l_] :=
           Rtetrad[i, j, k, l] = If[Intersection[A, combpos[i, j, k, l]] ≠ {},
             Rlist[[Position[A, Intersection[A, combpos[i, j, k, l]][[1]]][[1, 1]]]],
             -Rlist[[Position[A, Intersection[A, combneg[i, j, k, l]][[1]]][[1, 1]]]]];
```

Figure 2.6: Code for calculation of Riemann tensor in tetrad frame

Results can be mathced with 78, 79 of [3].

```
In[88]:= Rtetrad[2, 3, 2, 3]
```

$$\text{Out[88]=} \quad \frac{M\,r\,\left(3\,a^2 - 2\,r^2 + 3\,a^2\,\mathrm{Cos}[2\,\theta]\right)}{2\,\left(r^2 + a^2\,\mathrm{Cos}[\theta]^2\right)^3}$$

Figure 2.7: $R_{(1)(2)(1)(2)}$

Here one must remember that temporal components has index 1 in code. 1, 2, 3 spatial indices are written as 2, 3, 4 indices in code. In similar manner Q and P can be calculated(see [3] 80 to 99).

```
In[29]:= (* This calculates component ijklm of covariant derivative of R1dddd,
         where m is differentiation index*)
         defcR[i_, j_, k_, l_, m_] := defcR[i, j, k, l, m] = If[i == j || k == l, 0, FullSimplify[
              D[R1dddd[[i, j, k, l]], BLcoord[[m]]] - Sum[GUdd[[n, m, i]] R1dddd[[n, j, k, l]] +
                  GUdd[[n, m, j]] R1dddd[[i, n, k, l]] + GUdd[[n, m, k]] R1dddd[[i, j, n, l]] +
                  GUdd[[n, m, l]] R1dddd[[i, j, k, n]], {n, 1, 4}], assmp]];

In[30]:= CA = {};(*CA will contain all independent {i,j,k,l,m} *)

In[31]:= Do[CA = Append[CA, Append[A[[i]], j]], {i, 1, Length[A]}, {j, 1, 4}];

In[32]:= coCA = {};

In[33]:= Do[coCA = Append[coCA, defcR[CA[[i]] /. List → Sequence]], {i, Length[CA]}];

In[34]:= corR[i_, j_, k_, l_, m_] :=
           corR[i, j, k, l, m] = If[Intersection[CA, combpos[i, j, k, l, m]] ≠ {},
             coCA[[Position[CA, Intersection[CA, combpos[i, j, k, l, m]][[1]]][[1, 1]]]],
             -coCA[[Position[CA, Intersection[CA, combneg[i, j, k, l, m]][[1]]][[1, 1]]]]];

In[35]:= (*This function will return Q in tetrad frame as a vector: As written in the paper*)
         Q[i_, j_, k_, l_] :=
           Q[i, j, k, l] = Q[j, i, l, k] = Q[k, l, i, j] = Q[l, k, j, i] = FullSimplify[
                Table[If[i == j || k == l, 0, Sum[corR[i1, i2, i3, i4, i5] ijac[[i5, m]] ijac[[i1, i]]
                    ijac[[i2, j]] ijac[[i3, k]] ijac[[i4, l]], {i1, 1, 4}, {i2, 1, 4},
                   {i3, 1, 4}, {i4, 1, 4}, {i5, 1, 4}]], {m, {2, 3, 4, 1}}], assmp];
```

Figure 2.8: Calculating Q

```
In[36]:= CCA = {};(*CCA will be filled with independent components of {i,j,k,l,m,n} *)

In[37]:= Do[CCA = Append[CCA, Append[CA[[i]], j]], {i, 1, Length[CA]}, {j, 1, 4}];

In[38]:= co2CA = {};

In[39]:= pos[i_, j_, k_, l_, m_, n_] :=
           pos[i, j, k, l, m, n] = If[Intersection[CCA, combpos[i, j, k, l, m, n]] ≠ {},
             {Position[CCA, Intersection[CCA, combpos[i, j, k, l, m, n]][[1]]][[1, 1]], 1},
             {Position[CCA, Intersection[CCA, combneg[i, j, k, l, m, n]][[1]]][[1, 1]], -1}];
```

Figure 2.9: Calculating P

```
In[40]:=
(* This calculates component ijklmn of double covariant derivative of R1dddd,
where m and n are succesive differentiation index*)
def2cR[i_, j_, k_, l_, m_, n_] := def2cR[i, j, k, l, m, n] =
   If[i == j || k == l, 0, FullSimplify[(D[corR[i, j, k, l, m], BLcoord[[n]]] -
        Sum[GUdd[[q, n, i]] corR[q, j, k, l, m] + GUdd[[q, n, j]] corR[i, q, k, l, m] +
          GUdd[[q, n, k]] corR[i, j, q, l, m] + GUdd[[q, n, l]] corR[i, j, k, q, m] +
          GUdd[[q, n, m]] corR[i, j, k, l, q], {q, 1, 4}]) /. θ → Pi / 2, assmp]];

In[41]:= cor2R[i_, j_, k_, l_, m_, n_] :=
  cor2R[i, j, k, l, m, n] = co2CA[[pos[i, j, k, l, m, n][[1]] ]] pos[i, j, k, l, m, n][[2]];

In[42]:= calc2R[i_, j_, k_, l_, m_, n_] :=
  calc2R[i, j, k, l, m, n] = If[pos[i, j, m, k, l, n][[1]] < pos[i, j, k, l, m, n][[1]] &&
     pos[i, j, l, m, k, n][[1]] < pos[i, j, k, l, m, n][[1]],
    -cor2R[i, j, m, k, l, n] - cor2R[i, j, l, m, k, n], def2cR[i, j, k, l, m, n]];

In[43]:= Do[co2CA = Append[co2CA, calc2R[CCA[[i]] /. List → Sequence]], {i, Length[CCA]}];

In[44]:= ijac1 = FullSimplify[ijac /. θ → Pi / 2, assmp];
(*P is calculated only on equatorial plane*)

In[45]:= (*This function will return P in tetrad frame as a vector: As written in the paper*)
P[i_, j_, k_, l_] := P[i, j, k, l] =
  P[j, i, l, k] = P[k, l, i, j] = P[l, k, j, i] = Table[If[i == j || k == l, 0, FullSimplify[
        Sum[(1 / 2) (cor2R[i1, i2, i3, i4, i5, i6] + cor2R[i1, i2, i3, i4, i6, i5])
          ijac1[[i1, i]] ijac1[[i2, j]] ijac1[[i3, k]] ijac1[[i4, l]] ijac1[[i5, m]]
          ijac1[[i6, n]], {i1, 1, 4}, {i2, 1, 4}, {i3, 1, 4}, {i4, 1, 4},
         {i5, 1, 4}, {i6, 1, 4}], assmp]], {n, {2, 3, 4, 1}}, {m, {2, 3, 4, 1}}]
```

Figure 2.10: Calculating P

Result can be matched:

```
In[92]:= MatrixForm[P[2, 3, 2, 3]]

Out[92]//MatrixForm=
```

$$\begin{pmatrix} \frac{3M\left(-5a^2+(9M-4r)r\right)}{r^7} & 0 & 0 & 0 \\ 0 & \frac{3M\left(5a^2+r(-2M+r)\right)}{r^7} & 0 & 0 \\ 0 & 0 & \frac{9M\left(a^2+r(-2M+r)\right)}{r^7} & \frac{9aM\sqrt{a^2+r(-2M+r)}}{r^7} \\ 0 & 0 & \frac{9aM\sqrt{a^2+r(-2M+r)}}{r^7} & -\frac{3M\left(-3a^2+Mr\right)}{r^7} \end{pmatrix}$$

Figure 2.11: $P_{(a)(b)(1)(2)(1)(2)}$

```
In[90]:= Q[2, 3, 2, 3]
Out[90]= {(3 M Sqrt[a^2 + r (-2 M + r)] (-24 a^2 r^2 + 8 r^4 + 4 a^2 (a^2 - 6 r^2) Cos[2 θ] + a^4 (3 + Cos[4 θ]))) /
         (8 (r^2 + a^2 Cos[θ]^2)^(9/2)), (3 a^2 M r (a^2 - 2 r^2 + a^2 Cos[2 θ]) Sin[2 θ]) / (r^2 + a^2 Cos[θ]^2)^(9/2), 0, 0}
```

Figure 2.12: $Q_{(a)(1)(2)(1)(2)}$

$\Lambda_a^{(a)}$ denotes components of the four vecotrs, used for construction of Fermi normal coordinates, in tetrad frame. Components $\widetilde{\Lambda}$(see 100 to 147 in [3]) are also calculated.

```
        Λ0 = {Sqrt[1 + B^2 / r^2], 0, 0, B / r}; (*B=L-a E*)
In[48]:= Λ1 = {-B Sin[Ψ] / r, Cos[Ψ], 0, -Sqrt[1 + B^2 / r^2] Sin[Ψ]};
In[49]:= Λ2 = {0, 0, 1, 0};
In[50]:= Λ3 = {B Cos[Ψ] / r, Sin[Ψ], 0, Sqrt[1 + B^2 / r^2] Cos[Ψ]};
In[51]:= Λtilde0 = Λ0;
In[52]:= Λtilde1 = Λ1 Cos[Ψ] + Λ3 Sin[Ψ];
In[53]:= Λtilde2 = Λ2;
In[54]:= Λtilde3 = Λ3 Cos[Ψ] - Λ1 Sin[Ψ];
In[55]:= Λ = Transpose[{Λ0, Λ1, Λ2, Λ3}];
In[56]:= Λtilde = Transpose[{Λtilde0, Λtilde1, Λtilde2, Λtilde3}];
In[57]:= IΛtilde = FullSimplify[Inverse[Λtilde]];
In[58]:= AAA = ( 1  0       0  0
                 0  Cos[Ψ]  0  -Sin[Ψ]
                 0  0       1  0
                 0  Sin[Ψ]  0  Cos[Ψ] );
In[59]:= mat = FullSimplify[Λ.AAA];
```

Figure 2.13: Λ and $\widetilde{\Lambda}$

Results

Tidal component of gravitational potential is given as(see 129 of [3]):

$$\phi_{tidal} = \frac{1}{2}C_{ij}x^i x^j + \frac{1}{6}C_{ijk}x^i x^j x^k + \frac{1}{24}[C_{ijkl} + 4C_{(ij}C_{kl)} - 4B(kl|n|B_{ij)n}]x^i x^j x^k x^l + O(x^5) \quad (3.1)$$

Where i,j,k,l are summed over spatial components and*
$C_{ij} = R_{0i0j}$, $C_{ijk} = R_{0(i|0|j;k)}$, $C_{ijkl} = R_{0(i|0|j;kl)}$, $B_{ijk} = R_{k(ij)0}$

In tilde frame, assuming r $\gg$ M(>a):

$$\phi_{tidal} = \frac{1}{2}\widetilde{C}_{ij}\widetilde{x}^i\widetilde{x}^j + \frac{1}{6}\widetilde{C}_{ijk}\widetilde{x}^i\widetilde{x}^j\widetilde{x}^k + \frac{1}{24}[\widetilde{C}_{ijkl} + 4\widetilde{C}_{(ij}\widetilde{C}_{kl)} - 4\widetilde{B}(kl|n|\widetilde{B}_{ij)n}]\widetilde{x}^i\widetilde{x}^j\widetilde{x}^k\widetilde{x}^l + O(\widetilde{x}^5) \quad (3.2)$$

Code for calculating $\widetilde{C}$ & $\widetilde{B}$ components is given below:

*Where first bracket means averaged over permutations.
$A_{(p_1 p_2 \ldots |q_1| \ldots |q_2| \ldots p_n)} = \frac{1}{n!}\sum_{permutations} A_{permutations\ of\ p\ with\ q\ fixed.}$

```
In[60]:= (*Riemann tensor in tilde frame*)
calRtilde[i_, j_, k_, l_] := calRtilde[i, j, k, l] = If[i == j || k == l, 0,
    FullSimplify[Sum[(Rtetrad[i1, j1, k1, l1] /. θ → Pi / 2) mat[[i1, i]] mat[[j1, j]]
      mat[[k1, k]] mat[[l1, l]], {i1, 1, 4}, {j1, 1, 4}, {k1, 1, 4}, {l1, 1, 4}]]];

In[61]:= Rtildelist = {};

In[62]:= Do[Rtildelist = Append[Rtildelist, calRtilde[A[[i]] /. List → Sequence]], {i, Length[A]}];

In[63]:= Rtilde[i_, j_, k_, l_] := Rtilde[i, j, k, l] = If[Intersection[A, combpos[i, j, k, l]] ≠ {},
    Rtildelist[[Position[A, Intersection[A, combpos[i, j, k, l]][[1]]][[1, 1]]]],
    -Rtildelist[[Position[A, Intersection[A, combneg[i, j, k, l]][[1]]][[1, 1]]]]];

In[64]:= Ctilde[i_, j_] := Ctilde[i, j] = Rtilde[1, i, 1, j];

In[65]:= Btilde[i_, j_, k_] := Btilde[i, j, k] = (1 / 2) (Rtilde[k, i, j, 1] + Rtilde[k, j, i, 1]);

In[66]:= perm[ls_List] :=
  Module[{n = 0, htemp = {}, m = 0}, n = Length[ls]; htemp = Permutations[Range[n]];
   m = Length[htemp]; Table[ls[[htemp[[i, j]]]], {i, m}, {j, n}]]

In[67]:= QA = CA;

In[68]:= calQ[i_, j_, k_, l_, m_] :=
   calQ[i, j, k, l, m] = calQ[j, i, l, k, m] = calQ[k, l, i, j, m] = calQ[l, k, j, i, m] =
     If[i == j || k == l, 0, FullSimplify[Sum[(corR[i1, i2, i3, i4, i5] /. θ → Pi / 2)
         ijac1[[i5, m]] ijac1[[i1, i]] ijac1[[i2, j]] ijac1[[i3, k]] ijac1[[i4, l]],
        {i1, 1, 4}, {i2, 1, 4}, {i3, 1, 4}, {i4, 1, 4}, {i5, 1, 4}]]];
```

Figure 3.1: $\widetilde{C}_{ij}$, $\widetilde{C}_{ijk}$ & $\widetilde{B}_{ijk}$

Calculating $\widetilde{C}_{ijkl}$ needs extra care because we have to find a way to calculate sum of all combination of indices of double covariant derivative of Riemann tensor. The code is shown on next page.

```
In[69]:= coRtilde[i_, j_, k_, l_, m_] :=
          coRtilde[i, j, k, l, m] = coRtilde[j, i, l, k, m] = coRtilde[k, l, i, j, m] =
            coRtilde[l, k, j, i, m] = If[i == j || k == l, 0, FullSimplify[Sum[
                calQ[i1, j1, k1, l1, m1] mat[[i1, i]] mat[[j1, j]] mat[[k1, k]] mat[[l1, l]] mat[[
                  m1, m]], {i1, 1, 4}, {j1, 1, 4}, {k1, 1, 4}, {l1, 1, 4}, {m1, 1, 4}]]];

In[70]:= Ctilde[i_, j_, k_] := Ctilde[i, j, k] =
          Module[{temp = {}, len = 1, s = 0}, temp = perm[{i, j, k}]; len = Length[temp];
           Do[s = s + coRtilde[1, temp[[l, 1]], 1, temp[[l, 2]], temp[[l, 3]]], {l, len}];
           FullSimplify[s / len]];

In[71]:= calP[i_, j_, k_, l_, m_, n_] := calP[i, j, k, l, m, n] =
          calP[j, i, l, k, m, n] = calP[k, l, i, j, m, n] = calP[l, k, j, i, m, n] =
           If[i == j || k == l, 0, Module[{m1 = 1, n1 = 1}, m1 = Which[m == 1, 4, m > 1, m - 1];
             n1 = Which[n == 1, 4, n > 1, n - 1]; P[i, j, k, l, m1, n1]]];

In[72]:= co2Rtilde[i_, j_, k_, l_, m_, n_] :=
          co2Rtilde[i, j, k, l, m, n] = co2Rtilde[j, i, l, k, m, n] =
           co2Rtilde[k, l, i, j, m, n] = co2Rtilde[l, k, j, i, m, n] = If[i == j || k == l, 0,
              FullSimplify[Sum[(calP[i1, j1, k1, l1, m1, n1] mat[[i1, i]] mat[[j1, j]]
                  mat[[k1, k]] mat[[l1, l]] mat[[m1, m]] mat[[n1, n]]),
                {i1, 1, 4}, {j1, 1, 4}, {k1, 1, 4}, {l1, 1, 4}, {m1, 1, 4}, {n1, 1, 4}]]];

In[73]:= H[i_, j_, k_, l_, m_, n_] := H[i, j, k, l, m, n] = P[i, j, k, l][[m, n]];

In[74]:= Ctilde[i_, j_, k_, l_] := Ctilde[i, j, k, l] =
          Module[{temp = {}, len = 1, s = 0}, temp = perm[{i, j, k, l}]; len = Length[temp];
           Do[s = s + co2Rtilde[1, temp[[m, 1]], 1, temp[[m, 2]], temp[[m, 3]], temp[[m, 4]]],
            {m, len}]; FullSimplify[(s / len) /. P → H]];
```

Figure 3.2: $\widetilde{C}_{ijkl}$

```
In[75]:= Ctilde[4, 4, 4, 4]
```

$$\text{Out[75]=} \; -\frac{1}{r^9} 3\,M \left(3\,a^2\left(2\,B^2 + r^2\right) + 6\,a\,B\sqrt{1+\frac{B^2}{r^2}}\;r\sqrt{a^2 + r\,(-2\,M + r)} + r\left(3\,r^2\,(-2\,M + r) + B^2\,(-7\,M + 3\,r)\right)\right)$$

Figure 3.3: $\widetilde{C}_{3333}$: Matches with 160 of [3]

With these $\widetilde{C}$ & $\widetilde{B}$ we get (3.2). As shown:

```
In[76]:= xtilde = {y0, y1, y2, y3};

In[78]:= ϕ1tidal = FullSimplify[Sum[
           (1 / 2) (Ctilde[i, j] /. B → 0) xtilde[[i]] xtilde[[j]], {i, {2, 3, 4}}, {j, {2, 3, 4}}]];

In[79]:= ϕ2tidal =
          FullSimplify[Sum[(1 / 6) (Ctilde[i, j, k] /. B → 0) xtilde[[i]] xtilde[[j]] xtilde[[k]],
             {i, {2, 3, 4}}, {j, {2, 3, 4}}, {k, {2, 3, 4}}]] /. Sqrt[a^2 + r (-2 M + r)] → r;

In[80]:= ϕ31tidal =
          FullSimplify[Expand[FullSimplify[Sum[(1 / 24) (Ctilde[i, j, k, l] /. B → 0) xtilde[[i]]
                 xtilde[[j]] xtilde[[k]] xtilde[[l]], {i, {2, 3, 4}},
                {j, {2, 3, 4}}, {k, {2, 3, 4}}, {l, {2, 3, 4}}]] /. a → 0] /. M^2 → 0];

In[81]:= Cproduct[i_, j_, k_, l_] :=
          Cproduct[i, j, k, l] = Module[{sums = 0, pp = {}, len = 1}, pp = perm[{i, j, k, l}];
            len = Length[pp]; (1 / len) Do[sums = sums + (Ctilde[pp[[m, 1]], pp[[m, 2]]]
                  Ctilde[pp[[m, 3]], pp[[m, 4]]] /. B → 0), {m, len}] ; sums];

In[82]:= ϕ32tidal = FullSimplify[
            Sum[(1 / 6) Cproduct[i, j, k, l] xtilde[[i]] xtilde[[j]] xtilde[[k]] xtilde[[l]],
             {i, {2, 3, 4}}, {j, {2, 3, 4}}, {k, {2, 3, 4}}, {l, {2, 3, 4}}]] /. M^2 → 0;

In[83]:= Bsum[i_, j_, k_, l_] := Sum[Btilde[i, j, n] Btilde[k, l, n] /. B → 0, {n, {2, 3, 4}}]

In[84]:= Bproduct[i_, j_, k_, l_] := Bproduct[i, j, k, l] =
          Module[{sums = 0, pp = {}, len = 1}, pp = perm[{i, j, k, l}]; len = Length[pp]; (1 / len)
            Do[sums = sums + Bsum[pp[[m, 1]], pp[[m, 2]], pp[[m, 3]], pp[[m, 4]]], {m, len}] ; sums]

In[85]:= ϕ33tidal = FullSimplify[
            -Sum[(1 / 6) Bproduct[i, j, k, l] xtilde[[i]] xtilde[[j]] xtilde[[k]] xtilde[[l]],
              {i, {2, 3, 4}}, {j, {2, 3, 4}}, {k, {2, 3, 4}}, {l, {2, 3, 4}}]] /. M^2 → 0;

In[86]:= ϕtidal = ϕ1tidal + ϕ2tidal + ϕ31tidal + ϕ32tidal + ϕ33tidal
```

$$\text{Out[86]=}\ \frac{M\left(-2\,y1^2+y2^2+y3^2\right)}{2\,r^3}+\frac{M\,y1\left(2\,y1^2-3\left(y2^2+y3^2\right)\right)}{2\,r^4}-\frac{M\left(8\,y1^4-24\,y1^2\left(y2^2+y3^2\right)+3\left(y2^2+y3^2\right)^2\right)}{8\,r^5}$$

Figure 3.4: ϕ_{tidal}

Here, $\widetilde{x}$ is written as y. Code reproduces result 165 of Ishii's paper [3].

Thus, through Mathematica we have been able to calculate tidal potential in Fermi normal coordinate system in Kerr spacetime. This result can be used for further calculations.

Conclusions

The tidal potential that we have finally got is*:

$$
\begin{aligned}
\phi_{tidal} = \frac{M}{2r^3}[-2(\widetilde{x}^1)^2 + (\widetilde{x}^2)^2 + (\widetilde{x}^3)^2] \\
- \frac{M}{2r^4}\widetilde{x}^1[-2(\widetilde{x}^1)^2 + 3\{(\widetilde{x}^2)^2 + (\widetilde{x}^3)^2\}] \\
- \frac{M}{8r^5}[8(\widetilde{x}^1)^4 + 3(\widetilde{x}^2)^4 + 3(\widetilde{x}^3)^4 - 24\{(\widetilde{x}^1)^2(\widetilde{x}^2)^2 \\
+ (\widetilde{x}^1)^2(\widetilde{x}^3)^2\} + 6(\widetilde{x}^2)^2(\widetilde{x}^3)^2]
\end{aligned}
\tag{4.1}
$$

As noted in [3], this matches with Newtonian tidal potential from a point source of mass M at a distance r as $-\frac{M}{\sqrt{(\widetilde{x}^1+r)^2+(\widetilde{x}^2)^2+(\widetilde{x}^3)^2}}$.
To conclude :

- We see that, Fermi normal coordinates indeed serves as a useful frame as calculation involved gets simplified there.
- Though solution of geodesic equation, consequently the choice of Fermi normal coordinates are themselves dependent on spin parameter a, in Newtonian limit ϕ_{tidal} does not depend on a. Thus a freely falling observer locally can not distinguish between spinning and non-spinning star unless a and M is large, as expected.
- This result can be used in slow rotations also i.e. when a is a function of t but it can be, for practical purposes, assumed to be a constant.

*See 165 of [3]

Future Prospects

The whole work done throughout summer is a part of a bigger project which will run throughout semester I of academic year 2016-17. This part is mainly comprised of learning relevant theory and Mathematica. The ultimate goal of the project is to calculate tidal disruption rate of stars near naked singularity and I hope I will get some really useful and interesting results at the end. Beside this, in future, I wish to study physics near a black hole more rigorously. Also, if I get any chance, I wish to explore origin of gravitational waves near a black hole and using computaional tools I would like to build a simulation for that.

Acknowledgements

I sincerely express gratitude to my mentor Dr. Tapobrata Sarkar for giving me this valuable opportunity to learn under him. He has provided immeasurable amount of support and guidance throghout the project. I would like to thank my friend Mr. Amartya Saha with whom I frequently had discussions on theoritical topics. These discussions led me to clearer understanding of the subject. I also thank Mr. Ravindra Kumar Verma, Ph.D. student of our department, for helping me with LaTeX. Finally, I pay respect to my parents and elder sister for keeping me motivated throughout the project.

References

[1] Matthias Blau. *Lecture notes on general relativity*. December 15, 2015. http://www.blau.itp.unibe.ch/newlecturesGR.pdf

[2] I.S. Sokolnikoff. *Tensor Analysis Theory and Applications*. Applied Mathematics Series. NEW YORK. JOHN WILEY & SONS, INC. LONDON. CHAPMAN & HALL, LIMITED, 1951.

[3] M. Ishii, M. Shibata and Y. Mino, Phys. Rev. D **71**, 044017 (2005).

[4] Thomas A. Moore. *A General Relativity Workbook*. University Science Books. Mill Valley, California.

[5] A good discussion on energy-momentum tensor can be found at http://www2.warwick.ac.uk/fac/sci/physics/current/teach/module_home/px436/notes/lecture6.pdf.

[6] F.K. Manasse and C.W. Misner. Journal of Mathematical Physics **4**, 735 (1963).

[7] J.A. Marck. Proceedings of the Royal Soceity of London. Series A, Mathematical and Physical Sciences, Vol. 385, No. 1789 (Fb. 8, 1983), pp. 431-438.

[8] James M. Bardeen, William H. Press and Saul A. Teukolsky. The Astrophysical Journal, **178**:347-369, 1972.

[9] Mathematica tutorial at https://www.wolfram.com/learningcenter/tutorialcollection/CoreLanguage/CoreLanguage.pdf.

[10] Roozbeh Hazrat. *Mathematica: A Problem Centered Approach*. Springer.

[11] The complete Mathematica code is available at https://drive.google.com/file/d/0B3yQpROHCwaRTmJ4bUVxSGJSdHM/view?usp=sharing.